

報 告

数式処理ソフト Maple を活用した数学教材の作成

Development of Mathematics Teaching Materials
Using a Computer Algebra System中 村 俊 子*・玉 城 義 将**
NAKAMURA, Toshiko ; TAMASHIRO, Yoshimasa

概要：Maple は数値や数式の計算を行ったり，グラフやアニメーションを作成することのできる，プログラム言語を備えた数式処理ソフトである。第 14 巻第 1 号において，数学科 1 年次で学習する線形代数学と微分積分学の内容の Maple を活用した Web 教材の開発を報告したが，教材は Maple 7 を想定したものであり，現在本学 SCNL のシステムに導入されている Maple 10 には対応していない。そこで，Maple 10 で使用できるよう改訂を行い，さらに，「多変数関数の微分積分」や「微分方程式」等の解析学分野の教材を新たに作成・追加した。

1. はじめに

本報告では，カナダの Waterloo 大学の研究成果を元に開発された数式処理ソフト Maple を活用した数学科教材を作成したので紹介する。Maple を用いると数値，数式，図形の処理を統一的に扱うことが可能で，紙と鉛筆で行う計算をコンピュータで行ったり，グラフや図形，アニメーションを作成したり，数値計算を行うことができる。これにより，早く正確に計算結果を得ることができ，さらに，定理や公式，計算結果などを具体的かつ視覚的に理解・考察することができるようになる。Maple は数学の学習や研究において便利な道具である。また，数学以外の学問分野においても，科学や技術の計算やシミュレーションに広く応用されている。

城西情報科学研究第 14 巻第 1 号において，本学理学部数学科 1 年次の必修科目である「線形代数学 I」と「微分積分学 I」の内容を中心とする Maple を用いた教材を開発し，自習教材としてインターネット上で活用できるよう HTML 言語を用いて Web ページにまとめたことを報告した(参考文献(3))。しかしながら，これは当時，本学の情報教育システム(SCNL)に導入されていた Maple 7 を想定したものであり，2008 年 2 月現在インストールされている Maple 10 には対応していない。そこで，Maple 10 の基本的な操作方法について説明するとともに，Maple 10 で使

* 城西大学理学部数学科

** 城西大学大学院理学研究科数学専攻修士課程

用できるよう教材を改訂した。さらに、多変数関数の微分積分（第2次偏導関数と2変数関数の極値、条件付き極値、曲面）や微分方程式（単独方程式、連立方程式、アニメーション（シミュレーション））など、解析学分野の教材を新たに作成・追加し、Web教材の充実を図った。

尚、Web教材はURL <http://math.josai.ac.jp/~toshiko/maple/maple.html> において公開しており、数学科ホームページからもリンクされている。

2. Maple 10 を用いた Web 教材

目次

1章	Maple 10 とは
2章	Maple の起動と終了
3章	基本的な計算
4章	代入計算
5章	多項式の計算
6章	代数方程式 1 (1変数の方程式)
7章	代数方程式 2 (連立方程式)
8章	ベクトル計算
9章	行列計算
10章	微分積分の計算
11章	グラフの作成
12章	印刷
13章	テイラー展開
14章	平面曲線・空間曲線
15章	フーリエ級数
16章	2変数関数のグラフ
17章	第2次偏導関数と2変数関数の極値
18章	陰関数
19章	条件付き極値
20章	曲面
21章	微分方程式 1 (単独方程式)
22章	微分方程式 2 (連立方程式)
23章	アニメーション

1章 Maple 10 とは

Maple 10（以下、Maple と呼ぶ）とは、1980年代前半にカナダの Waterloo 大学で開発された、プログラミング言語を備えた数式処理ソフトです。Maple を使うと、紙と鉛筆で行う数学の計算や作図をコンピュータで行うことができます。

WWW 上の検索機能を活用して、Maple に関する情報を得ることができます。

実習 1.1 YAHOO のホームページ (<http://www.yahoo.co.jp>) の検索機能を活用して、数式処理に関するホームページを探してみましょう。

2章 Maple の起動と終了

城西大学 SCNL のシステムでは、Z!Stream にログインして、[スタート] → [プログラム] → [Maple 10] → [Maple 10] を選択すると、Maple が起動します。例えば、ワークシートのプロンプト > に続けて

```
> sin(Pi/3)
```

と入力し Enter キーを押すと、 $\sin \frac{\pi}{3}$ の値 $\frac{\sqrt{3}}{2}$ が表示されます。結果が近似的な数値でなく $\frac{\sqrt{3}}{2}$ のように厳密な値で出力されることが数式処理の特徴です。近似的な数値（浮動小数点近似）で表示するには、

```
> evalf(%)
```

もしくは

```
> evalf(sin(Pi/3))
```

とします。%は直前の結果（ここでは、 $\frac{\sqrt{3}}{2}$ ）を表します。

Maple を終了するには、[ファイル] → [終了] とします。ワークシートをファイルに保存する場合にはファイル名を入力します。

実習 2.1 Maple の起動と終了を行ってみましょう。

3章 基本的な計算

四則演算子は足し算「+」、引き算「-」、掛け算「*」、割り算「/」と入力します。また、べき乗「^（または**）および階乗「!」と入力します。ただし、ディスプレイ上では「*」は「・」、 2^3 は「2³」などと表示されます。例えば、 $(1.5+3.2) \times 4 - 5.2$ や $3x \div 4x^2$ は

```

> (1.5+3.2)*4-5.2
> 3x/(4x^2)

```

などと入力します。丸括弧でくくられている部分は通常の数学記号と同様に他の式より先に評価されます。

実習 3.1 次を計算してみましょう。4.0 は 4. と入力しても構いません。

- (1) $3+14-5$ (2) 32^{10} (3) $50!$ (4) $4\div 5$ (5) $4.0\div 5.0$ (6) $1\div 3+1\div 5$
 (7) $1\div 3+1\div 5$ の近似値 (8) $(6+5\times 3)\div 3$ (9) $3^{0.5}$ (10) 3^{3^3}

Maple では、数学辞典に載っているような基本的な関数は組み込み関数として用意されています。例えば、指数関数 e^x 「exp(x)」, 対数関数 $\log x$ 「log(x) または ln(x)」, 無理関数 \sqrt{x} 「sqrt(x)」があります。組み込み関数名がわからないときなどには、メニュー [ヘルプ] を選択して、調べることができます。Maple の操作に関するほとんどの情報はヘルプから得ることができます。

実習 3.2 必要があればヘルプを活用しながら、次を計算してみましょう。虚数単位 i は I と入力します。

- (1) $e^{2.5}$ (2) $\sqrt{2}+\sqrt{8}$ (3) $\sqrt{2}+\sqrt{8}$ の数値 (4) $\cos^2 1+\sin^2 1$
 (5) $\cos^2 1+\sin^2 1$ の数値 (6) $\cos^2 1.0+\sin^2 1.0$ (7) $(1+i)^{10}$

4章 代入計算

文字 x に値を代入することもできます。以下では、Maple が出力する部分 ($y:=x+2.4$ など) も併せて書いてあります。

```

> y:=x+2.4                                     (x+2.4をyとおく)
> subs(x=5,y)                                 y:=x+2.4
                                              (x=5を代入したときのyの値を求める)
> restart                                     7.4
                                              (変数x,yをクリア)

```

次のように関数を定義して、代入計算することもできます。

> f: = x - > x+2.4	f: = x - > x+2.4	(関数 f(x) = x+2.4 を定義)
> f(5)		(f(5) を計算)
> restart	7.4	(関数 f(x) をクリア)

変数は使い終わったら restart コマンドを用いて、クリア（それまで記憶されている内容を初期化）する習慣をつけておくとよいでしょう。

実習 4.1 関数 $f(x) = 2x^3 - 3x^2 + 5x + 1$ に対して、 $f(1)$, $f(-3)$, $f(0.2)$ の値をそれぞれ求めてみましょう。

5 章 多項式の計算

多項式を因数分解するには factor コマンドを用い、展開するには expand コマンドを用います。Maple を活用して、多項式の計算を行いましょう。

> expand((1-x)^3)	$1 - 3x + 3x^2 - x^3$	((1-x) ³ を展開)
> factor(%)	$-(x-1)^3$	(前の結果を因数分解)

実習 5.1 次を展開してみましょう。

(1) $(x+1)(x+2)$ (2) $(x^2+x-1)(x^2+x+1)$ (3) $(x+y)^{10}$

実習 5.2 次を因数分解してみましょう。

(1) x^2+3x-4 (2) $x^4+4x^3-x^2-16x-12$ (3) x^9-1 (4) $x^{10}-y^{10}$

6 章 代数方程式 1 (1 変数の方程式)

代数方程式の解を求めるには、solve コマンドを用います。

> solve(a*x^2+b*x+c = 0, x)	($ax^2+bx+c = 0$ の解を求める)
-----------------------------	---------------------------

2 次方程式の解の公式が表示されます。4 次までならば必ず厳密解を得ることができます。

5 次以上の場合、一般に代数的には解けないことが知られています。

```

> solve(x^2-6*x-3 = 0, x)          3+2√3, 3-2√3      (x^2-6x-3 = 0 の解を求める)
> evalf(%)                          (直前の結果の近似値を求める)
6.464101616, -4.64101616
> solve(x^5-x = -1, x)             (x^5-x = -1 の解を求める)
RootOf(z^5-z+1,index = 1), RootOf(z^5-z+1,index = 2),
RootOf(z^5-z+1,index = 3), RootOf(z^5-z+1,index = 4),
RootOf(z^5-z+1,index = 5)
> evalf(%)                          (直前の結果の近似値を求める)
0.7648844336+0.3524715460*I, -0.1812324445+1.083954101*I,
-1.167303978, -.1812324445-1.083954101*I,
0.7648844336-0.3524715460*I

```

“RootOf (多項式)” は “多項式 = 0 の解” を指しています。インデックス付けされた形で解が出力されています。

実習 6.1 次の方程式の解を求めてみましょう。また、解の近似値を求めてみましょう。

$$(1) x^2 - x - 1 = 0 \quad (2) x^4 - x^2 + 1 = 0 \quad (3) x^4 - x + 1 = 0$$

実習 6.2 3次方程式の解の公式 (Cardano の公式) を求めてみましょう。

7章 代数方程式 2 (連立方程式)

連立1次方程式

$$\begin{cases} x+2y+4z = 2 \\ 2x+3y+5z = 3 \\ 5x-4y-8z = 10 \end{cases}$$

を解くには、次のように入力します。

```

> solve({x+2*y+4*z = 2, 2*x+3*y+5*z = 3, 5*x-4*y-8*z = 10}, {x, y, z})
x = 2, y = -2, z = 1

```

実習 7.1 次の連立方程式の解を求めてみましょう。また解が求められない問題に関しては、その理由を考えてみましょう。

$$(1) \begin{cases} x+2y = 1 \\ 3x+5y = 4 \end{cases} \quad (2) \begin{cases} x+2y = 3 \\ 2x+4y = 6 \end{cases} \quad (3) \begin{cases} x+2y = 3 \\ 2x+4y = 7 \end{cases}$$

$$(4) \begin{cases} 2x+2y+z=2 \\ x-y+2z=-2 \\ 3x-2y-3z=1 \end{cases} \quad (5) \begin{cases} x+2y+4z=1 \\ 3x-5y+z=4 \\ 4x-y+2z=2 \end{cases} \quad (6) \begin{cases} 4x+2y+z-u=5 \\ x+2y-3z+u=7 \\ 3x-y+z+u=8 \\ -x-y+z+u=4 \end{cases}$$

* 連立1次方程式は、(係数行列の行列式) $\neq 0$ ならば、ただ1組の解を持ちます。

8章 ベクトル計算

Maple を活用して線形代数演算を行うときは、まず

```
> with(linalg):
```

として、パッケージ linalg をロードしておきます。パッケージ linalg の詳細については、ヘルプを参照して調べることができます。例を挙げて、ベクトルの演算方法について述べます。

```
> a := vector([1,2])           (ベクトル a = (1,2) の作成)
> b := vector([-3,2])         (ベクトル b = (-3,2) の作成)
> evalm(2*(a-2*b)-(3*a+b))    (2(a-2b)-(3a+b) の計算)
> innerprod(a,b)              (内積 a・b の計算)
> innerprod(vector([x,y,z]), vector([u,v,w])) (内積(x,y,z)・(u,v,w) の計算)
```

実習 8.1 $a = (1, 1, 0)$, $b = (-2, 1, 1)$, $c = (-1, 5, 1)$ とするとき、次を計算してみましょう。

$$(1) 5(a+b)-3(2a+c) \quad (2) (a+b) \cdot c \quad (3) 2(a-2b) \cdot (2a+c)$$

パッケージ LinearAlgebra をロードしても計算することができます。このときには、

```
> with(LinearAlgebra):
```

とした後で、

```
> a := Vector([1,2])           (ベクトル a = (1,2) の作成)
> b := Vector([-3,2])         (ベクトル b = (-3,2) の作成)
> 2*(a-2*b)-(3*a+b)           (2(a-2b)-(3a+b) の計算)
> a.b                          (内積 a・b の計算)
> Vector([x,y,z]). Vector([u,v,w]) (内積(x,y,z)・(u,v,w) の計算)
```


また、行列 A, B に対し、スカラー倍・和・差・積の計算は次のようにします。

> evalm(3*A)	(3A の計算)
> evalm(A+B)	(A+B の計算)
> evalm(A-B)	(A-B の計算)
> evalm(A& * B)	(AB の計算)

実習 9.1 次を計算してみましょう。

$$(1) \begin{pmatrix} 1 & -2 \\ 3 & -4 \end{pmatrix} + \begin{pmatrix} 0 & 5 \\ 3 & -2 \end{pmatrix} \quad (2) 4 \begin{pmatrix} 3 & 4 & 2 \\ -2 & 3 & -4 \end{pmatrix} - \begin{pmatrix} 0 & 1 & 1 \\ 1 & 3 & -2 \end{pmatrix} \quad (3) \begin{pmatrix} 2 & -2 \\ 3 & 4 \end{pmatrix} \begin{pmatrix} -3 & 1 \\ 2 & 2 \end{pmatrix}$$

$$(4) \begin{pmatrix} 1 \\ -2 \end{pmatrix} (-3 \ 4) \quad (5) (3 \ 2) \begin{pmatrix} 2 & 6 \\ -1 & -3 \end{pmatrix}$$

実習 9.2 次の行列 A, B について、以下を求めてみましょう。

$$A = \begin{pmatrix} 2 & 3 \\ 3 & 1 \end{pmatrix}, B = \begin{pmatrix} 3 & 4 \\ 5 & 3 \end{pmatrix}$$

$$(1) A^{-1} \quad (2) {}^t(A) \quad (3) (AB)^{-1} \quad (4) A^{-1}B^{-1} \quad (5) B^{-1}A^{-1}$$

$$(6) (B-A)^{-1} \quad (7) \det(B-A) \quad (8) \det(A) \quad (9) \det({}^tA)$$

*一般に、 ${}^t({}^tA) = A$, $(AB)^{-1} = B^{-1}A^{-1}$, $\det({}^tA) = \det(A)$ が成立します。また、行列 A に逆行列が存在するためには、 $\det(A) \neq 0$ であることが必要かつ十分条件です。

Maple 10 においては、

> with(LinearAlgebra):

と、パッケージ LinearAlgebra をロードして計算することもできます。

> A:= Matrix([[1,4,5],[2,5,8],[3,6,9]])	(3×3 行列 A の作成)
> Determinant(A)	(A の行列式 A を求める)
> Transpose(A)	(A の転置行列 tA を求める)
> A^(-1)	(A の逆行列 A^{-1} を求める)

と計算します。

また、行列 A, B に対し、スカラー倍・和・差・積の計算は次のように計算します。

> 3*A	(3A の計算)
> A+B	(A+B の計算)
> A-B	(A-B の計算)
> A . B	(AB の計算)

実習 9.3 LinearAlgebra をロードして、実習 9.1 と実習 9.2 を計算してみましょう。

10 章 微積分の計算

Maple を用いて導関数を求めるには、diff コマンドを用います。

> diff(x^10,x)	10x ⁹	(x ¹⁰ の導関数を求める)
----------------	------------------	----------------------------

高階導関数を求めるときは、微分する変数を並べて入力するか \$n (n は微分する回数) と入力します。例えば、 x^{10} の 3 階導関数を求めるには

> diff(x^10, x, x, x)
> diff(x^10, x\$3)

などとします。

実習 10.1 次の関数を微分してみましょう。

- (1) $x^3 - 2x^2 - 9$ (2) $(4x^2 - 5)^6$ (3) $e^{f(x)}$ (4) $f(x)g(x)$ (5) $f(x) \div g(x)$
 (6) $\sin(x^2)$ (2 階微分) (7) a^x (4 階微分) (8) $f(x)g(x)$ (5 階微分)

積分結果を得るためには、int コマンドを用います。

> int(x^n, x)	($\int x^n dx$ を求める)
> int(sin(Pi*x), x = 0 .. 1/2)	($\int_0^{1/2} \sin \pi x dx$ を求める)
> int(exp(-x^2), x = -infinity .. infinity)	($\int_{-\infty}^{\infty} e^{-x^2} dx$ を求める)

Maple では、無限大 ∞ は infinity と表します。

実習 10.2 次の不定積分を求めてみましょう。

$$(1) \int e^x dx \quad (2) \int (1 \div \sin t) dx \quad (3) \int a^x dx \quad (4) \int (1 \div (1+x^2)) dx$$

実習 10.3 次の定積分を求めてみましょう。

$$(1) \int_0^1 (2x+1)e^x dx \quad (2) \int_{-1}^1 1 \div (1+x^2) dx \quad (3) \int_a^b (x-a)(x-b) dx$$

11章 グラフの作成

関数 $y = \sin x$ のグラフを定義域 $[-2\pi, 2\pi]$ で描くには、次のように入力します。

```
> plot(sin(x), x = -2*Pi .. 2*Pi)
```

2つのグラフを同時に表示するには、次のように式を2つ並べて

```
> plot([sin(x), cos(x)], x = -2*Pi .. 2*Pi)
```

とします。Maple では、複数のグラフは自動的に曲線の色を変えて表示されます。白黒できれいに印刷するためには、次のように入力するとよいです。

```
> plot([sin(x), cos(x)], x = -2*Pi .. 2*Pi, linestyle = [1, 2], color = black)
```

線の色を黒 (color=black)、前者のグラフを実線で後者を破線 (linestyle=[1, 2]) で表示します。linestyle の値が 0 または 1 で実線、2 以上でさまざまな破線を表します。詳細については、ヘルプを参照してください。

実習 11.1 次の関数のグラフを描いてみましょう。

$$(1) y = x \cos(\pi x) \quad (\text{定義域 } [-20, 20]) \quad (2) y = x \sin(1 \div x) \quad (\text{定義域 } (0, 0.2])$$

12章 印刷

Maple を用いて得た結果をプリンターで印刷するときは、[ファイル] → [印刷] を選択してから、[OK] をクリックします。

用紙サイズはデフォルトでは、「レターサイズ」に設定されています。[ファイル] → [印刷の設定] を選択して、用紙サイズを「A4 サイズ」に変更しておきましょう。

実習 12.1 関数 $y = x$, $y = x^2$, $y = x^3$, $y = x^4$, $y = x^5$ (定義域はすべて $[0, 1]$ とする) のグラフを描いて、プリンターで印刷してみましょう。

13 章 テイラー展開

点 a を含む区間上で定義された関数 $f(x)$ が C^{n+1} 級 ($f, f', f'', \dots, f^{(n+1)}$ が連続) であるとき、次が成り立ちます ([テイラーの定理])。

$$f(x) = f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 + \dots + \frac{f^{(n)}(a)}{n!}(x-a)^n + R_{n+1}$$

$$R_{n+1} = \frac{f^{(n+1)}(c)}{(n+1)!}(x-a)^{n+1} \quad (\text{ここで } c \text{ は } a \text{ と } x \text{ の間の数})$$

点 a における n 次テイラー多項式

$$p_n(x) = f(a) + f'(a)(x-a) + \dots + \frac{f^{(n)}(a)}{n!}(x-a)^n$$

のグラフを描いてみましょう。例えば、 $f(x) = e^x$ の 0 における 9 次のテイラー多項式を考えてみます。Maple では、テイラー展開を求める `taylor` 関数が組み込み関数として用意されています。

```
> taylor(exp(x), x = 0, 10)
```

と入力すると、 $f(x) = e^x$ に対し、 $a = 0$, $n = 9$ の場合の展開式を求めることができます。得られた結果から剰余項 R_{10} を取り除いて、9 次のテイラー多項式 p_9 を取り出すには、`convert` コマンドにより、

```
> p9 := convert(%, polynomial)
```

とします。 $y = p_9(x)$ と $y = e^x$ のグラフを、例えば定義域 $[0, 3]$ として、並べて描いてみましょう。

```
> plot({exp(x), p9}, x = 0..3)
```

実習 13.1 e^x の 0 におけるテイラー多項式 p_3 , p_6 , p_9 を求め、 $y = e^x$ と一緒にグラフを描いて e^x への収束性を調べてみましょう。

実習 13.2 $1/(1-x)$ の 0 におけるテイラー多項式 p_3 , p_6 , p_9 を求め、 $y = 1/(1-x)$ と一緒に

グラフを描いて、 $1/(1-x)$ への収束性を調べてみましょう。

14 章 平面曲線・空間曲線

平面曲線 $P(t) = (\sin t, \cos t)$, $t \in [0, 2\pi]$ を表示するには、次のように入力します。

```
> plot([sin(t), cos(t), t = 0..2*Pi])
```

曲線が表示される領域の縦と横の比を同じにするには、次のように入力します。

```
> plot([sin(t), cos(t), t = 0..2*Pi], scaling = CONSTRAINED)
```

次に、空間曲線 $P(t) = (t, t^2, t^3)$, $t \in [-1, 1]$ を表示するには、

```
> with(plots):
> spacecurve([t, t^2, t^3], t = -1..1)
```

と入力します。

デフォルトでは座標軸等は表示されません。座標軸を表示するには、

```
> spacecurve([t, t^2, t^3], t = -1..1, axes = BOXED)
```

と入力します。他にも上の `axes=BOXED` を、`axes=NORMAL` もしくは `axes=FRAME` など、オプションとして指定することもできます。画面上ドラッグすることにより、空間曲線を様々な角度から見ることができます。

最後に曲線を滑らかに表示するには、

```
> spacecurve([t, t^2, t^3], t = -1..1, numpoints = n)
```

(n は十分大きな自然数) と入力します。

それでは以下の実習 14.1~14.6 の曲線を表示してみましょう。

実習 14.1 懸垂線 $P(t) = (t, (e^t + e^{-t})/2)$, $t \in [-1, 1]$

実習 14.2 リマソン $P(t) = ((\cos t + a)\cos t, (\cos t + a)\sin t)$, $t \in [0, 2\pi]$

- (1) $a = 1.5$ のとき (2) $a = 1.0$ のとき (3) $a = 0.5$ のとき

実習 14.3 $P(t) = (\tan t, 1/\cos t)$, $t \in [-\pi/6, \pi/6]$

実習 14.4 アステロイド $P(t) = (\cos^3 t, \sin^3 t)$, $t \in [0, 2\pi]$

実習 14.5 $P(t) = (e^t \cos t, e^t \sin t)$, $t \in [0, 30]$

実習 14.6 (1) $P(t) = (\cos t, \sin t, t)$, $t \in [0, 6\pi]$

(2) $P(t) = (t \cos t, t \sin t, t)$, $t \in [0, 6\pi]$

平面曲線 $\Gamma: P(t) = (x(t), y(t))$, $t \in [a, b]$ の長さは

$$\int_a^b \sqrt{x'(t)^2 + y'(t)^2} dt$$

空間曲線 $\Gamma: P(t) = (x(t), y(t), z(t))$, $t \in [a, b]$ の長さは

$$\int_a^b \sqrt{x'(t)^2 + y'(t)^2 + z'(t)^2} dt$$

(いずれも, $\int_a^b \|P'(t)\| dt$ の形) で与えられます。この公式を用いて曲線の長さを計算してみましょう。微分・積分については 10 章を参考にしましょう。

実習 14.7 前の実習 14.4~14.6 の曲線の長さを求めてみましょう。必要があれば、

> simplify(式)

として簡単な形で値を求めてみましょう。

15 章 フーリエ級数

区間 $[-\pi, \pi]$ 上で定義された関数 $f(x)$ に対し、

$$(a_0/2) + \sum_{n=1}^{\infty} (a_n \cos nx + b_n \sin nx)$$

を $f(x)$ のフーリエ級数といいます。ここで、 a_n, b_n は

$$a_n = (1/\pi) \int_{-\pi}^{\pi} f(x) \cos nx \, dx \quad (n = 0, 1, 2, \dots)$$

$$b_n = (1/\pi) \int_{-\pi}^{\pi} f(x) \sin nx \, dx \quad (n = 1, 2, 3, \dots)$$

により定まる数で、 $f(x)$ のフーリエ係数と呼ばれます。 $f(x) = x^2$, $x \in [-\pi, \pi]$ を例にとって、部分和

$$S_m(x) = s(x, m) = (a_0/2) + \sum_{n=1}^m (a_n \cos nx + b_n \sin nx)$$

が収束していく様子を観察してみましょう。フーリエ級数の定義にもとづいて

```
> f := x -> x^2
> a := n -> (1/Pi)*(int(f(x)*cos(n*x), x = -Pi..Pi))
> b := n -> (1/Pi)*(int(f(x)*sin(n*x), x = -Pi..Pi))
> s := (x,m) -> a(0)/2+sum(a(n)*cos(n*x)+b(n)*sin(n*x),n = 1..m)
```

と入力していきます。部分和 $s(x, 1)$ を求めるには、

```
> s(x, 1)
```

と入力します。部分和 $s(x, 1)$ のグラフを定義域 $[-\pi, \pi]$ で表示するには、

```
> plot(s(x, 1), x = -Pi..Pi)
```

とくに、 $f(x)$ のグラフと並べて表示するには、

```
> plot({f(x), s(x, 1)}, x = -Pi..Pi)
```

とします。以下の通り入力すると、 $f(x)$, $s(x, 1)$, $s(x, 5)$, $s(x, 9)$, $s(x, 13)$ のグラフを並べて表示します。

```
> plot({f(x), s(x, 1), s(x, 5), s(x, 9), s(x, 13)}, x = -Pi..Pi)
```

実習 17.1 部分和 $S_5(x)$, $S_9(x)$, $S_{13}(x)$ を求め、そのグラフを描いてみましょう。

実習 17.2 関数 $f(x) = -1(x \in [-\pi, 0]$ のとき), $0(x = 0$ のとき), $1(x \in [0, \pi]$ のとき) のフーリエ級数の部分 and $S_1(x)$, $S_5(x)$, $S_9(x)$, $S_{13}(x)$ を求め, そのグラフを描いてみましょう。ここで Maple では, 階段関数 $f(x)$ は `signum(x)` と表します。

実習 17.3 関数 $f(x) = |x|(x \in [-\pi, \pi])$ のフーリエ級数の部分 and $S_1(x)$, $S_5(x)$, $S_9(x)$, $S_{13}(x)$ を求め, そのグラフを描いてみましょう。ここで Maple では, $|x|$ は `abs(x)` と表します。

16 章 2 変数関数のグラフ

Maple を用いた 2 変数関数のグラフの描き方について説明します。デフォルトでは座標軸は表示されません。座標軸を表示するためには, はじめに

```
> with(plots):
> setoptions3d(axes = boxed)
```

と設定しておきましょう。 $z = 1 - x - y$ のグラフを定義域 $[-1, 1] \times [-1, 1]$ で表示するには, 次のように入力します。

```
> plot3d(1-x-y, x = -1..1, y = -1..1)
```

次のように入力すると画面上に等高線を表示させることもできます。

```
> plot3d(1-x-y, x = -1..1, y = -1..1, style = patchcontour)
```

等高線を細かく, 例えば 30 本引きたいときには

```
> plot3d(1-x-y, x = -1..1, y = -1..1, style = patchcontour,
contours = 30)
```

とします。描かれた図形をマウスでクリックして, ボタンを押したままドラッグすると, 画面を見る視点を変更することができます。真上や真横から見た図なども表示できます。

実習 16.1 関数 $z = x^3 + 3x^2 - y^2$ のグラフを描いてみましょう。

2つの関数のグラフ，例えば， $z = 1 - x - y$ と $z = 1 + x + y$ のグラフを同一空間に表示させるには，次のようにします。

```
> plot3d({1-x-y, 1+x+y}, x = -1..1, y = -1..1)
```

実習 16.2 関数 $z = x^3 + 3x^2 - y^2$ のグラフと次の点における接平面のグラフを，定義域を適当に定めて，同時に表示してみましょう。

- (1) $(-2, 0, 4)$ (2) $(1, 1, 3)$

* 関数 $z = f(x, y)$ のグラフ上の点 $(a, b, f(a, b))$ における接平面の方程式は，次の通りです。

$$z = f_x(a, b) \cdot (x - a) + f_y(a, b) \cdot (y - b) + f(a, b)$$

17章 第2次偏導関数と2変数関数の極値

1. 第2次偏導関数 2変数関数 $f(x, y)$ の第2次偏導関数，例えば $f_{xy}(x, y)$ を求めるには

```
> diff(f(x, y), x, y)
```

あるいは，次のようにします。

```
> D[1, 2](f)
```

実習 17.1 $\Delta f = 0$ を満たす関数 $f(x, y)$ を調和関数といいます。ここで， $\Delta f = f_{xx} + f_{yy}$ です (Δ はラプラシアンとよばれます)。次の関数が調和関数であるかどうか調べなさい。

- (1) $f(x, y) = x^2 - y^2$ (2) $f(x, y) = \log(x^2 + y^2)$

* 複雑な形の答えが出力された時には，simplify コマンドを用いて，より簡単な形で答えを表示できないか試してみましょう。

2. 極大・極小 関数 $f(x, y)$ において，ある $\delta > 0$ に対し

$$\sqrt{(x-a)^2 + (y-b)^2} < \delta, (x, y) \neq (a, b)$$

ならば，

$$f(x, y) < f(a, b) \quad (\text{あるいは, } f(x, y) > f(a, b))$$

が成り立つとき、 (a, b) を極大点（あるいは、極小点）といい、 $f(a, b)$ を f の極大値（あるいは、極小値）といいます。

関数 $f(x, y) = x^3 + y^3 - 3xy$ の極値を求めてみましょう。

```

> f := (x, y) -> x^3 + y^3 - 3*x*y
> solve({diff(f(x,y),x) = 0, diff(f(x,y),y) = 0}, {x,y}) (f の臨界点を求める)
      {y = 0, x = 0}, {y = 1, x = 1}, {……}
> evalf(%)
> det(H) := diff(f(x,y),x,x)*diff(f(x,y),y,y) - diff(f(x,y),x,y)^2
      (ヘッセ行列の行列式を def(H) とおく)
> subs(x = 0, y = 0, det(H)) ((0,0) における det(H) の値を計算)
      -9 (det(H) < 0 より, (0,0) で極値をとらない)
> subs(x = 1, y = 1, det(H)) ((1,1) における det(H) の値を計算)
      27
> subs(x = 1, y = 1, diff(f(x,y),x,x)) (fxx(1,1) を計算)
      6
      (det(H) > 0 かつ fxx(1,1) > 0 であるから (1,1) は極小点である)
> f(1,1) (極小値 f(1,1) を計算)
> plot3d(f(x,y), x = -0.5..1.5, y = -0.5..1.5, style = patchcontour,
contours = 30)
      ([-0.5, 1.5] × [-0.5, 1.5] における z = f(x,y) のグラフを表示)

```

注意(1) 行列 $H = \begin{pmatrix} f_{xx} & f_{xy} \\ f_{yx} & f_{yy} \end{pmatrix}$ はヘッセ行列とよばれます。

関数 $f(x, y)$ のヘッセ行列の行列式は、 $\det(H) = f_{xx}f_{yy} - f_{xy}f_{yx}$ です。特に、 f が C^2 級の場合は、 $f_{xy} = f_{yx}$ が成り立つから、 $\det(H) = f_{xx}f_{yy} - (f_{xy})^2$ となります。

注意(2) 関数 $f(x, y)$ が点 (a, b) で極値をとるならば、 $f_x(a, b) = 0$ かつ $f_y(a, b) = 0$ が成り立つ、すなわち (a, b) は臨界点です。ここで、 $f_x(a, b) = 0$ かつ $f_y(a, b) = 0$ を満たす点 (a, b) を臨界点といいます。臨界点 (a, b) において、

(i) $\det(H) > 0$ のとき、 $f_{xx} > 0$ ならば (a, b) は極小点、 $f_{xx} < 0$ ならば (a, b) は極大点

(ii) $\det(H) < 0$ のとき、 f は極大にも極小にもならない

ことが証明できます。(ii)の場合、 (a, b) は鞍点とよばれます。

実習 17.2 関数 $f(x, y) = (x^2 + y^2)^2 - 2x^2 + 2y^2$ の極値を求めなさい。また、関数のグラフを等高線をつけて表示し、結果を確かめなさい。

実習 17.3 関数 $f(x, y) = e^{-x^2-y^2}(2x^2+y^2)$ について以下の問いに答えよ。

- (1) 臨界点を求めなさい。
- (2) 極値を求めなさい。
- (3) $z = f(x, y)$ のグラフを作成して(1), (2)の結果を確認しなさい。

18 章 陰関数

Maple を用いて, $x^2 - xy + y^2 - 3 = 0$ で定まる陰関数 $y = \phi(x)$ のグラフを描いてみましょう。

```
> solve(x^2-x*y+y^2-3=0, y)
```

とする ($x^2 - xy + y^2 - 3 = 0$ を y について解く) と, $x^2 - xy + y^2 - 3 = 0$ で定まる陰関数が求まります。2 次方程式に対する解の公式を用いて手計算を行っても求められます。

$$\frac{1}{2}x + \frac{1}{2}\sqrt{-3x^2 + 12}, \frac{1}{2}x - \frac{1}{2}\sqrt{-3x^2 + 12}$$

これらのグラフを定義域 $[-2, 2]$ で描くには

```
> plot({1/2*x+1/2*sqrt(-3*x^2+12), 1/2*x-1/2*sqrt(-3*x^2+12)},
x = -2..2)
```

と入力します。

実習 18.1 $x^3 + 3x^2 - y^2 = 0$ で定まる陰関数のグラフを描いてみましょう。

Maple には, `implicitplot` という関数が用意されており, これを用いると簡単に表示することができます。

```
> with(plots):
> implicitplot(x^2-x*y+y^2-3=0, x = -2..2, y = -2..2)
```

曲線を滑らかに表示したいときには, `numpoint=10000` などとオプションをつけるとよいです。

実習 18.2 $x^3 + 3x^2 - y^2 = -1$, $x^3 + 3x^2 - y^2 = 0$, $x^3 + 3x^2 - y^2 = 1$ で定まる陰関数のグラフをそれぞれ描いてみましょう。また, $z = x^3 + 3x^2 - y^2$ のグラフを等高線を表示させて描いて比較してみましょう。

実習 18.3 曲線 $x^3+3xy+y^3=0$ を xy 平面上に図示してみましょう。また、 $z = x^3-3xy+y^3$ のグラフを等高線を表示させて描いて比較してみましょう。

19 章 条件付き極値

student パッケージの `extrema` コマンドを用いると、条件付き極値を求めることができます。例えば、条件 $g(x, y) = x^2 + (y^2/4) = 1$ の下での関数 $f(x, y) = xy$ の極値を求めるには次のように入力します。

```
> with(student):
> extrema(x*y, x^2+y^2/4 = 1, {x, y}, 's')           (条件付き極値を計算)
> s                                                   (極値点の候補を表示)
> allvalues(%)
```

注意 `extrema` (極値を求めたい関数, 制約条件, 変数, 's') の形で入力します。

出力結果はラグランジュの乗数法を用いて計算されており、極値点の候補は `s` に入れられています。

等位集合 $g(x, y) = 1$ の近傍上での $z = f(x, y)$ のグラフを表示して、上で得た結果の詳細を調べてみましょう。

```
> plot3d(piecewise(x^2+y^2/4 > 0.9 and x^2+y^2/4 < 1.1, xy, none), x = -2..2,
y = -2..2, grid = [100, 100])
```

注意 関数 $f(x, y) = \text{piecewise}$ (条件, $a(x, y)$, $b(x, y)$) は

$$f(x, y) = \begin{cases} a(x, y), & (x, y) \text{ が条件を満たすとき} \\ b(x, y), & \text{その他のとき} \end{cases}$$

という関数を表します。

実習 19.1 関数 $f(x, y) = e^{-x^2-y^2}(2x^2+y^2)$ の極値を求めなさい。

実習 19.2 関数 $f(x, y) = e^{-(x+y-2)^2}$ と $g(x, y) = x^3 - 3xy + y^3$ について、以下の問いに答えなさい。

- (1) 条件 $g(x, y) = 0$ の下での $f(x, y)$ の極値を求めなさい。
- (2) 条件 $f(x, y) = 1$ の下での $g(x, y)$ の極値を求めなさい。

20 章 曲 面

パラメータ表示された曲面

$$S : (x(s, t), y(s, t), z(s, t)), s \in [a, b], t \in [c, d]$$

を表示してみましょう。例えば、球面 $S : (\sin s \cos t, \sin s \sin t, \cos s)$, $s \in [0, \pi], t \in [0, 2\pi]$ を表示するには、次のように入力します。

```
> plot3d([sin(s)*cos(t), sin(s)*sin(t), cos(s)], s = 0..Pi, t = 0..2Pi)
```

実習 20.1 次の曲面を表示しなさい。

- (1) $(2 \cos s, 2 \sin s, t)$, $s \in [0, 2\pi], t \in [-1, 1]$
- (2) メビウスの輪
 $((2+t \cos s/2)\cos s, (2+t \cos s/2)\sin s, t \sin s/2)$, $s \in [0, 2\pi], t \in [-1, 1]$
- (3) トーラス
 $((2 + \cos t)\cos s, (2 + \cos t)\sin s, \sin t)$, $s \in [0, 2\pi], t \in [0, 2\pi]$

$y = f(x) (a \leq x \leq b)$ のグラフを x 軸の回りに回転してできる回転体において、回転面は

$$(s, f(s)\cos t, f(s)\sin t), s \in [a, b], t \in [0, 2\pi]$$

とパラメータ表示できます。また

$$\begin{aligned} \text{体積} &= \pi \int_a^b \{f(x)\}^2 dx \\ \text{回転面の面積} &= 2\pi \int_a^b \sqrt{1 + \{f'(x)\}^2} |f(x)| dx \end{aligned}$$

が成り立ちます。

実習 20.2 $y = \sin x (0 \leq x \leq 3\pi)$ のグラフを x 軸の回りに回転してできる回転体について、次の問いに答えなさい。

- (1) 回転面を表示しなさい。
- (2) 体積を求めなさい。
- (3) 表面積を求めなさい。

* Maple では $|x|$ は `abs(x)` と表します。

2 つ以上の曲面を同時に表示してみましょう。

```

> with(plots):
> A := plot3d([sin(s)*cos(t), sin(s)*sin(t), cos(s)], s = 0..Pi, t = 0..0.5*Pi)
      (球面  $x^2+y^2+z^2=1$  上で「 $x > 0$  かつ  $y > 0$  の部分」を表示)
> B := plot3d([sin(s)*cos(t), sin(s)*sin(t), cos(s)], s = 0..Pi, t = Pi..1.5*Pi)
      (球面  $x^2+y^2+z^2=1$  上で「 $x < 0$  かつ  $y < 0$  の部分」を表示)
> display({A,B})

```

実習 20.3 次を表示しなさい。

- (1) 球面 $x^2+y^2+z^2=1$ 上で「 $y > 0$ かつ $z > 0$ の部分」と「 $y < 0$ かつ $z < 0$ の部分」
- (2) 球面 $x^2+y^2+z^2=1$ 上で「 $x > 0$ かつ $y > 0$ かつ $z > 0$ の部分」と「 $x > 0$ かつ $y < 0$ かつ $z < 0$ の部分」と「 $x < 0$ かつ $y > 0$ かつ $z < 0$ の部分」と「 $x < 0$ かつ $y < 0$ かつ $z > 0$ の部分」

実習 20.4 懸垂線 $y = (e^x + e^{-x})/2$, $x \in [-1, 1]$ を x 軸の回りに回転してできる懸垂面を表示しなさい。また、その表面積を求めなさい。

実習 20.5 実習 20.1(3)のトーラスは円 $(x-2)^2+z^2=1$ (2つの関数 $x = 2 \pm \sqrt{(1-z^2)}$ のグラフ) を z 軸の回りに回転してできる回転体です。すなわち、 $y = 2 \pm \sqrt{(1-x^2)}$ のグラフを x 軸の回りに回転してできる回転体と合同である。次の問いに答えなさい。

- (1) 体積を求めなさい。
- (2) 表面積を求めなさい。

21 章 微分方程式 (単独方程式)

1. 等速運動, 等加速度運動

x 軸上を一定の速度 1 で動く動点 P の時刻 t における位置 (x 座標) を $x(t)$ と表すと, $x'(t) = 1$ が成り立ちます。 $x'(t) = 1$ を満たす関数 $x(t)$ を求めてみましょう。

```
> dsolve(diff(x(t),t) = 1,x(t))
```

特に、初期条件 $x(0) = 0$ を満たす関数を求めるときには、

```
> dsolve({diff(x(t),t) = 1, x(0) = 0}, x(t))
```

とします。また、 x 軸上を一定の加速度 -1 で運動する動点 P の時刻 t における位置 (x 座標) を $x(t)$ と表すと、 $x''(t) = -1$ が成り立ちます。 $x(t)$ を求めるには次のようにします。

```
> dsolve(diff(x(t), t, t) = -1, x(t))          (微分方程式  $x''(t) = -1$  を解く)
> dsolve({diff(x(t), t, t) = -1, x(0) = 0, D(x)(0) = 1}, x(t))
      (微分方程式  $x''(t) = -1$  を初期条件  $x(0) = 0, x'(0) = 1$  の下で解く)
```

注意 微分方程式を解くには、`dsolve (微分方程式, $x(t)$)` あるいは `dsolve ({微分方程式, 初期条件}, $x(t)$)` などと入力します。

実習 21.1 次の微分方程式の初期値問題を解きなさい。

$$(1) v'(t) = -g, v(0) = v_0 \quad (2) x''(t) = -g, x(0) = x_0, x'(0) = v_0$$

2. 増殖と減衰

バクテリアが容器に入っていて、一定の増殖率 k で増殖するとします。このとき、時刻 t におけるバクテリアの量を $x(t)$ で表すと、

$$x' = kx \quad (x'(t) = kx(t))$$

が成り立ちます。また、増殖率が $k-x$, $(k-x)(x-\lambda)$ であるときには、

$$x' = (k-x)x, \quad x' = (k-x)(x-\lambda)x$$

がそれぞれ成り立ちます。パッケージ DEtools の DEplot コマンドを用いると、これらの微分方程式の解曲線を図示することができます。

```

> with(DEtools) :
> de1 := diff(x(t), t) = x(t)           (微分方程式  $x' = x$  を de1 とおく)
> DEplot(de1, x(t), t = -5..5, x = 0..2)
    (微分方程式 de1 の解曲線を  $-5 \leq t \leq 5, 0 \leq x \leq 2$  の範囲に表示)
> init1 := {[0,0], [0,0.5], [0,1]}
    (初期条件  $x(0) = 0$  と  $x'(0) = 0, 0.5, 1$  を指定)
> DEplot(de1, x(t), t = -5..5, x = 0..2, init1)
    (微分方程式 de1 の初期条件 init1 を満たす解曲線を表示)

```

実習 21.2 次の微分方程式の初期値問題の解曲線を表示せよ。

- (1) [ロジスティック方程式] $x' = (1-x)x$, 初期条件 $x(0) = 0, 0.3, 0.7, 1.0$
- (2) $x' = (1-x)x$ 初期条件 $x(0) = 0, 0.7, 1.0, 0.4, 0.6, 0.8, 1.01$

実習 21.3 [半減期] $x' = -kx$, $x(0) = a$ を満たす関数 $x(t)$ が $x(t) = a/2$ となるのは, $t = (\log 2)/k$ のときであることを示しなさい。

3. 振 動

単振動を記述する微分方程式 $x'' = -x$ の解を求め, そのグラフを表示してみましょう。

```

> dsolve(diff(x(t), t, t) = -x(t), x(t))
> dsolve({diff(x(t), t, t) = -x(t), x(0) = 1, D(x)(0) = 0}, x(t))
    x(t) = cos(t)
> plot(rhs(%), t = 0..50)           (直前の結果(%)の右辺のグラフを表示)

```

実習 21.4 [摩擦を伴う振動 (減退振動)]

- (1) $x'' = -x - kx'$ の解を $k = 1, 2, 3$ の場合に求めなさい。
- (2) $x'' = -x - kx'$, $x(0) = 1, x'(0) = 0$ の解を $k = 1, 2, 3$ の場合に求め, それらのグラフを表示せよ。

実習 21.5 [強制振動を伴う振動] $x'' = -x + \sin \omega t$, $x(0) = 0, x'(0) = 0$ の解を $\omega = 0, 0.2, 0.5, 0.9, 1, 1.5$ の場合に求め, それらのグラフを $0 \leq t \leq 500$ の範囲で表示せよ。

22 章 微分方程式 (連立方程式)

1. 生態系 (ロトカ・ボルテラ方程式)

ある 2 種類の生物種 X, Y の時刻 t における個体数がそれぞれ $x(t), y(t)$ と表せ、 x と y の変化が次の微分方程式 (系) に従うとしましょう。

$$\begin{cases} \frac{dx}{dt} = x(a_1 - b_1x + c_1y) \\ \frac{dy}{dt} = y(a_2 + b_2x - c_2y) \end{cases}$$

ここで、 a_1 と a_2 はそれぞれ X と Y の種内の増殖率、 b_1 と c_2 は種内の競争率とします。また、 c_1 と b_2 は種間相互作用から定まる定数とします。 c_1 と b_2 がともに正の値のとき方程式は「協調系」とよばれ、ともに負のとき「競争系」、 c_1 が正で b_2 が負のとき「捕食者・被食者系」とよべれます。例えば、 $a_1 = a_2 = b_2 = c_1 = 1, b_1 = c_2 = 0$ のとき、すなわち、

$$x' = x(1+y), y' = y(1+x)$$

の解曲線を xy 平面上に図示して、生態系の時間変化を調べてみましょう。

```
> with(DEtools):
> de1 := [diff(x(t), t) = x(t)*(1+y(t)), diff(y(t), t) = y(t)*(1+x(t))]
> init1 := seq([x(0) = 0.2*i, y(0) = 0.9-0.2*i], i = 1..4)
> DEplot(de1, [x(t), y(t)], t = 0..30, [ini1], x = 0..4, y = 0..4, stepsize = 0.1)
```

実習 22.1 ロトカ・ボルテカ方程式系において、次の場合に解曲線を図示しなさい。

- (1) $a_1 = a_2 = b_1 = c_1 = c_2 = 1$
- (2) $a_1 = a_2 = b_2 = c_1 = 1, b_1 = c_2 = -1$
- (3) $a_1 = a_2 = 1, b_1 = b_2 = c_1 = c_2 = -1$
- (4) $a_1 = a_2 = b_1 = b_2 = c_2 = -1, c_1 = 1$
- (5) $a_1 = b_2 = 1, b_1 = c_2 = 0, a_2 = c_1 = -1$

2. 熱対流 (ローレンツ方程式)

上下の平行な壁の間の液体に対し、下の壁の温度が上の壁の温度より十分大きいときに起こる対流運動はローレンツ方程式

$$\begin{cases} \frac{dx}{dt} = -\sigma(x-y) \\ \frac{dy}{dt} = -xz + rx - y \\ \frac{dz}{dt} = xy - bz \end{cases}$$

により記述されます。ここで、 $\sigma = 10$, $r = 28$, $b = 8/3$ のとき、カオス現象が観察されることが知られています。Maple を用いて、このときの初期条件 $x(0) = y(0) = z(0) = 10$ を満たす方程式の解の挙動を調べてみましょう。

```
> de2 := [diff(x(t),t) = -10*(x(t)-y(t)), diff(y(t),t) = -x(t)*z(t)
+28*x(t)-y(t), diff(z(t),t) = x(t)*y(t)-8/3*z(t)]
> DEplot3d(de2, {x(t), y(t), z(t)}, t = 0..100, [[x(0) = 10, y(0) = 10,
z(0) = 10]], scene = [x(t),y(t),z(t)], x = -50..50, y = -50..50, z = 0..50,
stepsize = 0.01)
```

注意 解は不規則に振舞うことが知られています。また、解の極限集合はローレンツアトラクターとよばれています。

実習 22.2 初期条件を変えてローレンツ方程式の解を求め図示せよ。

23 章 アニメーション

1. アニメーション

これまでの学習をもとに、さまざまなアニメーションを作成してみましょう。

```
> with(plots):
> animate(sin(x-t), x = 0..4*Pi, t = 0..10)
```

と入力し、描かれたグラフをクリックして表示されるメニューバー上のアニメーション再生ボタンをクリックすると、アニメーションが始まります。メニューバー上のボタンを使って、アニメーションの実行速度を変えたり 連続再生モードに設定することなどができます。

注意 $u(x, t) = \sin(x-t)$ は微分方程式 $u_{tt} = u_{xx}$ (波動方程式) を満たします。すなわち波動方程式の解です。

進行波解とよばれる微分方程式の解のシミュレーションを行ったり、らせんを一定の角速度で回転させてみましょう。

```

> animate(tanh(x-t), x = -20..20, t = 0..25)
                                     (進行波解のシミュレーション)
> animate([r*cos(r-t), r*sin(r-t)], r = 0..6*Pi, t = 0..10, numpoints = 100,
scaling = constrained)
                                     (回転するらせんのシミュレーション)
    
```

注意 (1) $u(x, t) = \tanh(x-t)$ は微分方程式 $u_t = u_{xx} + (2u-1)(1-u^2)$ を満たします。このように、ある関数 $\phi(y)$ と定数 c を用いて $u(x, t) = \phi(x-ct)$ と表せる一定波形、一定速度 c で進む解を進行波解といいます。数理生物学や燃焼理論への応用上も重要な解です。

(2) 曲線 $(r \cos r, r \sin r)$, $r \in [a, b]$ はアルキメデスのらせんとよべれます。例題の回転するらせんは、結晶成長を記述する数理モデルに現れる解です。

KdV (Korteweg-de Vries) 方程式 $u_t + 6uu_x + u_{xxx} = 0$ のソリトン解 (孤立波解)

$$u(x, t) = 2k^2 \operatorname{sech}^2(kx - 4k^3t)$$

(k は定数) や 2 ソリトン解

$$u(x, t) = (2 \log(1 + ce^{2kx-8k^3t} + de^{2hx-8h^3t} + cd(k-h)^2 e^{2kx-8k^3t+2hx-8h^3t} / (k+h)^2))_{xx}$$

(c, d, k, h は定数) のシミュレーションを行ってみましょう。

```

> animate(2*sech(x-4*t)^2, x = -25..25, t = -5..5, numpoints = 200)
                                     (ソリトン解 (k = 1 の場合) のシミュレーション)
> v := (x, t) -> 1 + exp(2*x-8*t) + exp(x-t) + exp(2*x-8*t+x-t)/9
> u := (x, t) -> 2*diff(log(v(x, t)), x, x)
> amimate(u(x, t), x = -30..30, t = -6..6, numpoints = 200)
                                     (ソリトン解 (c = d = k = 1, h = 1/2 の場合) のシミュレーション)
    
```

3. おわりに

本報告の中の教材の幾つかは、数学科 2 年次開講科目である「数式処理による解析 (実習を含む)」の授業の中で実際に使用している。授業では、黒板を使用しての理論的な内容の講義と Maple を

用いたコンピュータ実習を隔週で行い、数学の理論の定着を目指している。グラフィックスやシミュレーション作成を通して、学生の興味や関心を喚起することができ、また、数学の理論を具体的かつ視覚的に理解・考察させることができ、概ね好評を得ている。さらに、数理的思考力や数量・図形感覚の養成に役立てることができていると感じる。

今後、新たな題材の追加、教材開発の方向性の模索・検討を行い、教材をさらに充実・発展させていきたいと考えている。

参考文献

- (1) 榎アंक (2001年), HTMLタグ辞典, 翔泳社
- (2) 示野信一 (1999年), Maple Vで見る数学ワールド, シュプリンガー・フェアラーク東京
- (3) 中村俊子/坂田義晃 (2004年), Web上の数式処理ソフト Maple を用いた数学科教材サイトの構築, 城西情報科学研究, 第14巻第1号, 9-27
- (4) K. M. ヒール/M. L. ハンセン/K. M. リカード (示野信一・管野幸夫・西山享・橋本隆司・室正和訳) (1998年), Maple V リリース5 ラーニングガイド, シュプリンガー・フェアラーク東京

(Received Feb. 9, 2008)